
'These three lines change the configuration setting in the menu before uploading the code to the PIC
'They allow you to upload it without changing any of the settings.
#CONFIG
    __CONFIG _CONFIG1, _INTRC_IO & _PWRTE_ON & _MCLR_OFF & _LVP_OFF
#ENDCONFIG

' Set the internal oscillator frequency to 8 MHz
DEFINE OSC 8
OSCCON.4 = 1
OSCCON.5 = 1
OSCCON.6 = 1


'Turn off the A/D convertors (required for the PIC16F88, to use associated pins for digital I/O)
ANSEL = 0


'declare vars
'start button is a N0 button
startbutton var PORTB.0

'This is a reset line that allows the Arduino 
'to tell the PIC that it is time to reset for the next round
reset var PORTB.1 

'handshake connection that tell the Arduino that there is 
'money available when high
ready var PORTB.3
'handshake connection that tell the Arduino to start the round when high
running var PORTB.2

'input from photo transistor
sensor1 var PORTB.4       

'var that stores the current amount of money that is in the system
money var byte
'variable that stores the state that the menu screen should be in numerically
state var byte

'startup commands resets the menu state and the money value
 state = 0
 money = 0
  
  

mainloop:
  'count the money
   if(sensor1 == 0) then          'when coin goes through
     money = money + 25           'add 25 cents
     pause 1000                   'wait to let the coin roll past 
   endif

   if (reset == 0) then           'if Arduino is telling the PIC to reset
     state =  0                   'reset the menu to default
     low ready			  'set ready low to reset signals to Arduino
     low running		  'set running low so the Arduino doesn't start another round
     pause 500			  'pause to let Arduino finish its reset
   endif

  
   
   if(state != 2) then            'if not running
     if(money >= 25) then         'if money has at least 25 cents
       state = 1 		  'ready
       if(startbutton == 1) then  'if the startbutton is pressed
         state = 2 		  'running       
         money = money - 25	  'subtract money for round cost
       endif
     else
       state = 0 		  'waiting for money
     endif
   endif
    
   if(state == 1) then		  'if ready
     high ready			  'set ready signal to Arduino high
   endif
   
   if(state == 2) then		  'if running
     High running		  'set ready an d running high for the Arduino telling it to start
     high ready
   endif    
   
   if(state ==0) then		  'if waiting for money set outputs to Arduino low
     low ready
     low running
   endif         
   
   Goto mainloop   		  ' Do it forever
   End




